

AppBee-Mod & AppBee-SIP
Documentation for use with the BASIC Stamp[®]

Application Interface Boards for the
ZigBee/802.15.4 XBee[®] and XBee-PRO[®]
Wireless Network Modems

Version 0.8. September, 2006

BASIC Stamp is registered trademark of Parallax, Inc
XBee and XBee-PRO are registered trademarks of MaxStream

Document and boards are copyrighted 2006, SelmaWare Solutions

Overview.....	3
Device Connections and Data Communications.....	4
BASIC Stamp to XBee Using the AppBee Mod Board	5
PC to XBee Using AppBee Mod Board Using a USB2SER.....	6
BASIC Stamp to XBee Using the AppBee SIP Board	7
PC to XBee Using AppBee SIP Board Using a USB2SER.....	7
Network Communications and Configuration	7
Point-to-Point Messages	7
Broadcast Messages	9
Configuring the XBee from X-CTU Software.....	9
Configuring the XBee from Terminal.....	9
Summary of Common Configuration Commands	11
Configuring the XBee from Code.....	12
Communication Strategies	14
Accepting Data to the BASIC Stamp.....	14
Controlling and Polling Nodes with Codes	15
Using the BASIC Stamp for Fast Polling a Network of Devices	18
Reading XBee Information.....	19
Sending Direct Data with Sleeping.....	20
Sending Data to StampDAQ.....	22
Sending Data to StampPlot.....	23
Board Construction and Mounting.....	24
FAQs.....	24
Resource Links.....	26

Overview

The XBee is a wireless modem using the IEEE 802.15.4 Low-Rate Wireless Personal Area Network protocol (LR-WPAN) for Wireless Sensor Networks (WSN). This allows addressable communications between nodes. Data may be sent to individual nodes (point-to-point), or to all nodes in range using a broadcast address. The devices use clear channel assessment (CCA) on a CSMA/CA network, which helps ensure devices do not talk-over one another. In point-to-point communications, error checking, acknowledgements and retries are used to ensure data delivery. The use of flow control (RTS) helps ensure devices, such as the BASIC Stamp, do not miss incoming data.

AppBee-Mod is an application interface board that connects to the AppMod dual row 20-pin header of many Parallax BASIC Stamp® boards for power and quick communications to the BASIC Stamp. This board has connections to many I/O of the XBee to provide greater flexibility in using the various feature of the XBee.

AppBee-SIP is an application interface board with a 6-pin single row header for easy connection to breadboard and custom printed circuit boards. This board has only essential connection for power, communications and accessing sleep modes of the XBee, though additional jumpers and circuitry may be used to expand on the provided I/O.

Both boards provide a carrier for the XBee modules, 3.3V regulator, and 5V to 3.3V logic converters for safe interfacing. Both boards also support direct interfacing to a computer terminal program using Parallax's **USB2SER** USB to serial interface device. *All features of Maxstream's X-CTU software may be used with the USB2SER except reprogramming of the firmware.*

This document provides guidance on the use of the XBee with the BASIC Stamp but relies on the user to access the full XBee documentation to use the full features of this exciting device. Many other controllers may also use these interfaces boards, but this document will focus on BASIC Stamp use.


The following is summary of key XBee features from Maxstream's documentation for firmware version 1.083.

High Performance, Low Cost	Low Power
<p>XBee</p> <ul style="list-style-type: none"> • Indoor/Urban: up to 100' (30 m) • Outdoor line-of-sight: up to 300' (100 m) • Transmit Power: 1 mW (0 dBm) • Receiver Sensitivity: -92 dBm <p>XBee-PRO</p> <ul style="list-style-type: none"> • Indoor/Urban: up to 300' (100 m) • Outdoor line-of-sight: up to 1 mile (1500 m) • Transmit Power: 100 mW (20 dBm) EIRP • Receiver Sensitivity: -100 dBm <p>RF Data Rate: 250,000 bps</p>	<p>XBee</p> <ul style="list-style-type: none"> • TX Current: 45 mA (@3.3 V) • RX Current: 50 mA (@3.3 V) • Power-down Current: < 10 μA <p>XBee-PRO</p> <ul style="list-style-type: none"> • TX Current: 214 mA (@3.3 V) • RX Current: 55 mA (@3.3 V) • Power-down Current: < 10 μA
<p>Advanced Networking & Security</p> <hr/> <p>Retries and Acknowledgements</p> <p>DSSS (Direct Sequence Spread Spectrum)</p> <p>Each direct sequence channels has over 65,000 unique network addresses available</p> <p>Point-to-point, point-to-multipoint and peer-to-peer topologies supported</p> <p>Coordinator/End Device operations supported</p> <p>128-bit Encryption (downloadable firmware version coming soon)</p> <p>Transparent and API operation</p>	<p>Easy-to-use</p> <hr/> <p>No configuration necessary for out-of box RF communications</p> <p>Free X-CTU Software (Testing and configuration software)</p> <p>Small form factor</p> <p>Network compatible with other 802.15.4 devices</p> <p>AT and API Command Modes for configuring module parameters</p> <p>Free & Unlimited Technical Support</p>

NOTE: Firmware version 802.15.4 v1.083 is the recommended and discussed version for this document. Examples are not guaranteed to work with other firmware versions of the XBee. Please use the XBee documentation from MaxStream for full discussions on device use.

Device Connections and Data Communications


The XBee unit handles the packaging of data for transmission and error checking and acknowledgements on reception. All that is required is to send the device serial asynchronous data (9600, 8-N-1 by default) for transmission and accepting data addressed to that particular node. The XBee is a 3.3V device using non-inverted data (3.3V = Logic 1, 0V = Logic 0). Signal conditioning from RS-232 ports or from the 5V BASIC Stamp are required. For simple data flow all that is required is data sent to Din and accepted from Dout. RTS may be enabled so that the XBee does not send data it received via RF before the interfaced device is ready to accept the data (Firmware version 1.083 required).



Simple XBee data communications with interfaced device

BASIC Stamp to XBee Using the AppBee Mod Board

The AppBee Mod board plugs into the AppMod header on the Parallax Board of Education (BOE) and other common boards. To provide flexibility in pin assignments, jumper wires must be placed between the AppMod header 20-pin connector and XBee communications header as shown where P0 (0) of the BASIC Stamp is connected to RX, 2 to *TX and 6 to *RTS.



XBee header connections starting with * denote it is an input to the XBee and are signal conditioned to 3.3V through the 74LVC244A buffer. The BASIC Stamp header also has an additional row of solder connectors in the event the user wishes to have soldered connections between the XBee header pads and the BASIC Stamp header (Even P-numbers only).

A summary of XBee header I/O is as follows:


XBee Header Pin	Description
Vss	Vss or Ground of the BASIC Stamp and system.
RX	Received data from the XBee's Dout to the BASIC Stamp.
*TX	Data from the BASIC Stamp to the XBee's Din, buffered to 3.3V.

*Reset	Optional RESET input for the XBee. Connecting and bringing this line LOW (0V) will reset the XBee. Use of this line is recommended if the BASIC Stamp modifies baud rates or guard times so that on a BASIC Stamp reset the XBee can be reset by code prior to sending configuration information.
*RTS	Ready to Send – Buffered input to the XBee to indicate the BASIC Stamp is ready to receive data. Use SERIN with flow control to accept data. Highly recommended to ensure data is not missed. (XBee firmware 1.083 required)
*RSSI	Receiver signal strength indicator. A PWM output of the XBee, which can be configured to indicate the strength of the last, received RF data. This may be accomplished in command mode with serial data also.
ASSOC	XBee output indicating it has associated with a network under certain configurations. Please see XBee documentation. This can be read by the BASIC Stamp or used to drive an LED through a 100-ohm resistor.
SLP_I	XBee output, Sleep Indicate. When sleep modes are used, this output can be used to indicate the status of the device. This may be read by the BASIC Stamp or used to drive an LED through a 100-ohm resistor.
CTS	Output of the XBee. May be used for flow control to ensure the XBee is ready to receive new data.
*SLP_R	Buffered input to the XBee, Sleep Request. When configured, a HIGH on the pin will place the XBee in a low-power sleep mode (<10uA) conserving battery life. The XBee cannot send nor receive serial or RF data when sleeping.

NOTE: Be sure NOT to configure BASIC Stamp I/O as outputs to the outputs from the XBee. A 5V output of the BASIC Stamp and 3.3V output of the XBee may be damaging to either or both devices.

PC to XBee Using AppBee Mod Board Using a USB2SER


The XBee may communicate to a personal computer (PC) via USB using Parallax's USB2SER device. This allows full interface features using Maxstream's X-CTU features except the ability to download new firmware. **Note: The XBee board should not be connected to the BASIC Stamp on RX/TX/RESET while the USB2SER is connected.**



By supplying power to Vss and Vin (5V to 18V) of the AppBee Mod board header and using the USB2SER device, stand-alone use of the XBee with PC may be accomplished.

BASIC Stamp to XBee Using the AppBee SIP Board

The AppBee SIP board allows use the XBee Modem with a breadboard, or other 0.1" (2.54mm) spaced holes with a single 6-pin header. It provides 3.3V regulation and buffering from 5V devices. Vin should be in the range of 5V to 26V. With the BASIC Stamp, Vin of the supply is the preferred source due to XBee current draw.




The AppBee SIP provides connections for TX/RX/RTS and a Sleep request input to the XBee as discussed in the AppBee header discussion.

While the I/O is limited, connections to the breadboard may be made directly from the XBee module headers, for outputs such as Sleep indicate or inputs such as RESET. **Note: Be sure to condition any 5V inputs to the XBee to 3.3V or less through the use of voltage dividers or other devices.**

PC to XBee Using AppBee SIP Board Using a USB2SER

The XBee may communicate to a PC via USB using Parallax's USB2SER device. This allows full interface features using MaxStream X-CTU features, except the ability to download new firmware. **Note: The XBee should not be connected to the BASIC Stamp on RX/TX while the USB2SER is connected.**

By supplying power to Vss and Vin (5V to 18V) of the AppBee SIP board header and using the USB2SER device, stand-alone use of the XBee with PC may be accomplished.



Network Communications and Configuration

Point-to-Point Messages

Each node on the network is identified by a unique 16-bit address allowing over 65,000 devices on a single network. As illustrated, a device at address 0 may communicate with various addresses, or device node 1 may communicate directly with another node. This forms a point-to-point network where anyone node may communicate with another (range dependent).




Figure 2: Node Communication and Addressing

The XBee units have numerous settings for configuration, two of which are:

MY – The node's own address (16-bit)

DL – Destination Low (16-bit)

By configuring these, a node is given an address, and a node to communicate to. For example, for node 0 to send data to node 1:

MY = 0

DL = 1

The data will be delivered with error checking, an acknowledgement, and up to 3 retries if needed. For node 2 to send data to node 0:


MY = 1

DL = 0

Of course, node 1 may also send data to node A2 (hexadecimal) by setting:

DL = A2

The following figure shows a network of 5 XBee's using various monitoring and communications choices. Two are connected to the PC's to act as terminal programs, or to accept data into specialized software, such as StampPlot or StampDAQ.



Broadcast Messages


A broadcast message is one sent from a node to all listening nodes. This is performed without acknowledgements or retries. To send a broadcast message, DL is set to FFFF. This is useful if you wish to poll all devices or send a control message to all. When devices send data they use clear channel assessment (CCA) to help ensure they do not talk over one another.

DL = FFFF

Configuring the XBee from X-CTU Software

By default, all devices are assigned an address of 0 on power-up. This may be changed through either the X-CTU software Modem Configuration tab or using AT commands from a terminal window, or serially from a device, such as the BASIC Stamp.

This figure shows configuring from the X-CTU software using the USB2SER device. Once the device is READ, settings are modified and new settings downloaded using WRITE.




MaxStream's X-CTU Modem Configuration Software

Configuring the XBee from Terminal

The next figure shows the configuration using a terminal window set for 9600, 8-N-1. The device is placed in command mode by issuing a "+++". To ensure data that contains this does not shift the unit into command mode, a guard time before and after is required – that is, no other data can be sent for a time period before or after the +++, this includes carriage returns.

Once in command mode, AT commands are sent to configure the unit. ATCN is used to exit command mode. Sending the command with no value will cause the unit to return the current value or setting as is shown.

In this example, once in command mode the values of DL and MY are requested, then modified, then requested once again. Command mode is exited and data is sent between terminals.



The screenshot shows the X-CTU [COM10] terminal window. The interface includes tabs for PC Settings, Range Test, Terminal, and Modem Configuration. The Terminal tab is active, displaying a series of AT commands and responses. The status bar at the bottom indicates the port is COM10, the baud rate is 9600 8-N-1, and the flow control is NONE. The received data is 52 bytes.

```
+++OK
atmy
0
atd1
0

atmy 1
OK
atd1 A2
OK

atmy
1
atd1
A2

atcn
OK

hello
how are you?
Fine,. thanks!
LED = ON
```

Even though the X-CTU software is used for the example, HyperTerminal ® or any terminal program running at 9600 Baud, 8-N-1, may be used. Data may be sent and received through the terminal window as well.

Summary of Common Configuration Commands

A summary of common configuration commands is listed. Again, please see the XBee documentation for full discussions and many more options for use.

Command	Description
MY	Sets/ Read the current node address. Address are in hexadecimal, 0-FFFE ATMY ATMY A2
DL	Sets/ Read the destination node address. Addresses are in hexadecimal, 0-FFFE ATDL ATDL 1 Use FFFF to broadcast to all nodes.
CH	Sets/Reads the current RF channel in the 2.4GHz range. This can be used for frequency separation between different 802.15.4 networks or to limit interference from other 2.4GHz devices. ATCH ATCH A Please see XBee documentation for allowed values.
ID	Sets/Reads the current Personal Area Network ID (PAN ID). Allows network separation between devices in same location. ATID ATID 3350 Please see XBee documentation for allowed values.
NI	Sets/Reads the node descriptive identification, 20 characters. ATNI ATNI Temperature Sensor
BD	Sets/Reads the value for the current serial baud rate. 0 - 7 (standard baud rates) 0 = 1200 bps 1 = 2400 2 = 4800 3 = 9600 4 = 19200 5 = 38400 6 = 57600 7 = 115200 ATBD 2
D6	Sets/Reads the status of RTS flow control. Set to 1 to use RTS flow control. ATD6 ATD6 1
GT	Sets/Reads the value of guard times required before and after issuing +++ to enter command mode. Decrease this value to 3 to allow fast configuration changes, such as the need to quickly change destination addresses. Values are in hexadecimal, please see XBee documentation for a full discussion. ATGT ATGT 3
RO	Sets/Reads how long to wait for characters before packetizing and sending. Value is 3 by default, FF maximum. This is good to ensure a line of data from the BASIC Stamp is sent as a single packet instead of being broken up where data from other units may be interposed. ATRO FF
ND	Performs a network discovery – Allows viewing of all nodes, including addresses, MAC addresses, name identification, and RSSI level (dBm). ATND
DB	Reads the dBm level of the last reception in hexadecimal, please see XBee documentation for a full discussion. ATDB
SM	Sets/Reads the sleep mode of the XBee. While sleeping, power consumption is <10uA. When set to 1, the SLP_R input of the AppBee Mod board may be used to place the

	device in an idle state. ATSM ATSM 1
WR	Writes the current configuration to XBee non-volatile memory so that on power up the settings will persist. ATWR
RE	Restores the default settings of the XBee. ATRE
PL	Sets/Reads the current output power level of XBee-PRO Modems. Please see XBee documentation. ATPL 3
CN	Exits command mode. A timeout period will also exit from command mode. ATCN

Command instructions may also be issued in shorthand, such as:
ATMY 0, DL 0, D6 1, CN (and a carriage return)

Configuring the XBee from Code

The BASIC Stamp can use the SEROUT instruction to place the XBee in command mode and send configuration information. A guard time, sending +++, followed by a guard time is required. Once configured, ATCN is issued to exit command mode.

```
' {$STAMP BS2}
' {$PBASIC 2.5}
' *****
' * Config_XBee_Example.BS2 *
' * Illustrates configuring the XBee *
' * from the BASIC Stamp. *
' *****
myAddr    CON $1    ' Node Address in hex
DestAddr  CON $0    ' Destination address in hex

Baud      CON 84    ' Baud rate, 9600, 8-N-1, non-inverted, on BS2.

RX        PIN 0     ' Receive Pin
TX        PIN 2     ' Transmit Pin
RTS       PIN 6     ' Flow control Pin

HIGH TX           ' Place TX pin High (Idle state)

DEBUG CLS, "Configuring XBee..."
PAUSE 2000        ' Guard time for command sequence
SEROUT TX, Baud, ["+++"] ' Enter command mode
PAUSE 2000        ' Guard time for command sequence
SEROUT TX, Baud, ["ATNI BS2 Test Node", CR,
                  "ATMY ", HEX myAddr, CR,
                  "ATDL ", HEX DestAddr, CR,
                  "ATD6 1", CR,
                  "ATCN", CR] ' Set description
                               ' Set node address
                               ' Set destination address
                               ' Use RTS for flow control
                               ' Exit command mode

PAUSE 1000
DEBUG "Configuration Complete!", CR
DO                               ' Send message forever at 1S intervals
```

```
SEROUT TX, Baud,["Hello Node!", CR]  
PAUSE 1000  
Loop
```

Communication Strategies

These examples illustrate different communication and control strategies with the BASIC Stamp. The code assigns pin connections using PIN instructions and are configured for:

TX on P0
RX on P2
RTS on P6

Accepting Data to the BASIC Stamp

The BASIC Stamp can accept byte data very easily from the XBee using the SERIN instruction: SERIN RX\RTS, Baud, [ByteVariable]

Whether accepting data from another XBee equipped BASIC Stamp or from a terminal window, flow control should be used with the BASIC Stamp. Ensure the XBee is configured for RTS flow control by setting ATD6 1.

The following code will accept a byte of data to be displayed as a character. The node is set to address 1 (MY = 1), and the destination is set to 0. Data from the transmitting unit should have DL = 1 or DL = FFFF. Note that with a rapid string of characters, every other character will be missed due to RTS being slow in stopping the next byte from being sent from the XBee.

```
' {$STAMP BS2}
' {$PBASIC 2.5}
' *****
' * Accept_Character.bs2 *
' * Illustrates accepting a Byte *
' * recieved as a character *
' *****
myAddr    CON $1      ' Node Address
DestAddr  CON $0      ' Destination address

Baud      CON 84      ' Baud rate, 9600, 8-N-1, non-inverted, on BS2.

RX        PIN 0       ' Receive Pin
TX        PIN 2       ' Transmit Pin
RTS       PIN 6       ' Flow control Pin

RFin      VAR Byte

HIGH TX           ' Set TX pin to idle state

DEBUG CLS, "Configuring XBee..."
PAUSE 2000        ' Guard time for command sequence
SEROUT TX,Baud,["+++ "]
PAUSE 2000        ' Guard time for command sequence
SEROUT TX,Baud,["ATNI BS2 Test Node",CR,
                "ATMY ", HEX myAddr,CR,
                "ATDL ", HEX DestAddr,CR,
                "ATD6 1",CR,
                "ATCN",CR]
                ' Set description
                ' Set node address
                ' Set destination node address
                ' Use RTS for flow control
                ' Exit command mode
```

```

PAUSE 1000
DEBUG "Configuration Complete!",CR
DO
  SERIN RX\RTS,Baud,[RFin]           ' Accept and view as character
  DEBUG RFin
LOOP

```

A timeout on SERIN may also be used:

```

DO
  SERIN RX\RTS,Baud,100,Timeout,[RFin]   ' Accept and view as character
  DEBUG Rfin
Timeout:
  ' Other processing code
LOOP

```

Decimal values may also be accepted using the DEC Modifier. Data sent must end with a carriage return.

```

SERIN RX\RTS,Baud,[DEC RFin]           ' Accept and view as decimal value
DEBUG DEC RFin, CR

```

Controlling and Polling Nodes with Codes



Simple control action and data polling can be obtained. For example the following hardware, control action and reading of data can be performed:

- Servo on P12
- LED on P7
- Speaker on P1
- Photo Resistor and Capacitor (RC Network) on P10

From a terminal window (or BASIC Stamp) connected to an XBee configured to send data to a node at address 0:

- Send **d** to read just data
- Send **D** to read text + data (Requires string processing by BASIC Stamp)
- Send **0** to **9** to sound tones
- Send **n** (as in On) and **f** (as in off) to control the LED
- Send **L**, **R**, **C** to position the servo

With an array of such boards, each can be read and controlled individually by setting the DL to that boards address, or all at once using a broadcast (DL = FFFF).



```
' {$STAMP BS2}
' {$PBASIC 2.5}
' *****
' * Node_Control.bs2 *
' * Illustrates control and polling *
' * of nodes *
' *****
myAddr CON $1 ' Node Address
DestAddr CON $0 ' Destination address

Baud CON 84 ' Baud rate, 9600, 8-N-1, non-inverted, on BS2.

RX PIN 0 ' Receive Pin
TX PIN 2 ' Transmit Pin
RTS PIN 6 ' Flow control Pin
LED PIN 7
Servo PIN 12
Buzzer PIN 1

RC PIN 10

RFin VAR Byte
DataOut VAR Byte
x VAR Byte

HIGH TX ' Set TX pin to idle state

DEBUG CLS, "Configuring XBee..."
PAUSE 2000 ' Guard time for command sequence
SEROUT TX,Baud,["+++"] ' Enter command mode
```



```

PAUSE 2000 ' Guard time for command sequence
SEROUT TX,Baud,["ATNI Photo Node",CR, ' Set description
    "ATMY ", HEX myAddr,CR, ' Set node address
    "ATDL ", HEX DestAddr,CR, ' Set destination node address
    "ATD6 1",CR, ' Use RTS for flow control
    "ATCN",CR] ' Exit command mode

PAUSE 1000
DEBUG "Configuration Complete!",CR

DO
  GOSUB READ_Data
  GOSUB ReadnControl
LOOP

DEBUG CLS,"Ready",CR

READ_Data: ' Read RC Network
HIGH RC
PAUSE 5
RCTYPE RC,1,DataOut
RETURN

Send_Full: ' Tx data with text
DEBUG "Sending full text ", DEC DataOut," ", CR
SEROUT TX, Baud, [" Photo is: ", DEC DataOut,CR]
RETURN

Send_Value: ' TX data only
DEBUG "Sending value only ", DEC DataOut," ", CR
SEROUT TX, Baud, [DEC DataOut,CR]
RETURN

ReadnControl: ' Read incoming data and control
SERIN RX\RTS,Baud,250,Timeout2,[RFin]
SELECT RFin
CASE "D"
  GOSUB Send_Full
CASE "d"
  GOSUB Send_Value
CASE "N","n"
  DEBUG "LED ON ",CR
  HIGH LED
CASE "F","f"
  DEBUG "LED OFF ",CR
  LOW LED
CASE "L","l"
  DEBUG "Servo Left ",CR
  FOR x = 1 TO 50
    PULSOUT Servo, 500
    PAUSE 20
  NEXT
CASE "C","c"
  DEBUG "Servo Center ",CR
  FOR x = 1 TO 50
    PULSOUT Servo, 750
    PAUSE 20

```

```

NEXT
CASE "R","r"
  DEBUG "Servo Right          ",CR
  FOR x = 1 TO 50
    PULSOUT Servo, 1000
    PAUSE 20
  NEXT
CASE "1" TO "9"
  DEBUG "Sounding tone at ", DEC RFin-48 * 500,"Hz      ",CR
  FREQOUT buzzer, 500, RFin-48 * 500
ENDSELECT

TimeOut2:
RETURN

```

Using the BASIC Stamp for Fast Polling a Network of Devices

Given a number of boards with the previous template for control, a 'coordinator' BASIC Stamp may poll the devices. Ensure each node is given a unique address, such as 1 through 10 for a value of MyAddr, and that they all have node 0 as the destination address (DestAddr). Notice that the guard time (ATGT 3) is reduced in the configuration to allow the DL address to be quickly updated to poll devices quickly in a round-robin fashion. With a change in guard time, the XBee may need to be reset with either a power cycling or use of the RESET line to accept new configuration changes.

```

' {$STAMP BS2}
' {$PBASIC 2.5}
' *****
' * Polling_Coordinator.bs2          *
' * Illustrates polling a number     *
' * of BS2-XBee Nodes               *
' *****

myAddr    CON $0
DestAddr  VAR Word

LowAddr   CON 1          ' first to poll
HighAddr  CON 10         ' last to poll, A hex
Baud      CON 84

RX        PIN 0
TX        PIN 2
RTS       PIN 6

X         VAR Byte
DataIn    VAR Word

HIGH TX
DEBUG CLS, "Configuring XBee..."
PAUSE 2000
SEROUT TX,Baud,["+++"]
PAUSE 2000
SEROUT TX,Baud,["ATGT 3",CR]          ' **** minimal guard time
SEROUT TX,Baud,["ATNI Coordinator",CR]

```

```

SEROUT TX,Baud,["ATDL ", HEX DestAddr,CR]
SEROUT TX,Baud,["ATMY ", HEX myAddr,CR]
SEROUT TX,Baud,["ATD6 1",CR]
SEROUT TX,Baud,["ATCN",CR]
PAUSE 1000
DO
  FOR DestAddr = lowAddr TO highAddr          ' Cycle through addresses
  GOSUB SetDest                                ' Go set address
  SEROUT Tx, Baud, ["d"]                       ' Send d for raw data
  DEBUG CR, "Polling ", DEC DestAddr           ' show status to DEBUG
  SERIN Rx\RTS, Baud,1000, timeout, [DEC DataIn] ' Accept returned data with timeout
  DEBUG " Value:", DEC DataIn                  ' If data arrives, show it
  Timeout:
  NEXT                                          ' Check next
PAUSE 2000
LOOP                                           ' Repeat

SetDest:
  PAUSE 20
  SEROUT TX,Baud,["+++"]                       ' Enter command mode with minimal guard time
  PAUSE 2
  SEROUT TX,Baud,["ATDL", HEX DestAddr, ",CN",CR] ' Update DL for polling
RETURN

```

Reading XBee Information

The BASIC Stamp may also request information from the XBee, such as current address, Receiver Signal Strength Indicator dBm levels, and a host of other information discussed in the XBee Documentation. Reducing the guard time will increase the speed at which this information is received. All data is returned as hexadecimal values and displayed in decimal. Valid ranges are -40dBm to nearly -100dBm.

A routine, WaitOK, is used to accept the XBee's "OK" so that it does not interfere with actual data being received.

```

' {$STAMP BS2}
' {$PBASIC 2.5}
' *****
' * Get_RSSI.bs2          *
' * Illustrates accepting a Byte *
' * received as a character *
' * and reads/displays RSSI dBm *
' *****
myAddr    CON $1      ' Node Address
DestAddr  CON $0      ' Destination address

Baud      CON 84      ' Baud rate, 9600, 8-N-1, non-inverted, on BS2.

RX        PIN 0       ' Receive Pin
TX        PIN 2       ' Transmit Pin
RTS       PIN 6       ' Flow control Pin

RFin      VAR Byte
dBm       VAR Byte

```

HIGH TX

```
DEBUG CLS, "Configuring XBee..."
PAUSE 2000 ' Guard time for command sequence
SEROUT TX,Baud,["+++"] ' Enter command mode
PAUSE 2000 ' Guard time for command sequence
SEROUT TX,Baud,["ATNI BS2 Test Node",CR, ' Set description
    "ATMY ", HEX myAddr,CR, ' Set node address
    "ATDL ", HEX DestAddr,CR, ' Set destination node address
    "ATD6 1",CR, ' Use RTS for flow control
    "ATGT 3",CR, ' Reduce Gaurd Time
    "ATCN",CR] ' Exit command mode

PAUSE 1000
GOSUB WaitOK ' Accept OK data
DEBUG "Configuration Complete!",CR
DO ' Accept and view a character
    SERIN RX\RTS,Baud,1000,Timeout,[RFin]
    DEBUG RFin
    GOSUB GetRSSI
Timeout:
LOOP

GetRSSI: ' Read RSSI data
PAUSE 20 ' Guard time
SEROUT TX, Baud, ["+++"] ' Command mode
PAUSE 2
SEROUT TX, Baud, ["ATDB",CR] ' Request level
SERIN RX\RTS, Baud, 1000, Timeout2, [HEX dBm] ' Accept data
DEBUG " dbm: -", DEC dBm ' Display data
Timeout2:
SEROUT TX, Baud,["ATCN",CR] ' Exit command mode
GOSUB WaitOK ' Accept OK from XBee
DEBUG CR
RETURN

WaitOK:
SERIN RX\RTS, Baud, 100,Timeout3, [RFin] ' Accept OK data
GOTO WaitOK
Timeout3:
Return
```

Sending Direct Data with Sleeping

Of course, the unit does not need to be polled to send data. A simple SEROUT will transmit the data to the intended recipient at the DL address. Since the XBee uses the IEEE 802.15.4 protocol and CSMA/CA (like home wireless networking routers), data delivery without conflict from other units is nearly guaranteed.

This sample program will send data (a value of X), while putting both the XBee and BASIC Stamp asleep between transmissions. The SLP_R pin of the AppBee board should be connected to P8. SLP_I on the AppBee-MOD, or going directly to the XBee connector header (pin 13), the user may add a sleep indicator LED via a resistor to their breadboard.

```

'{$STAMP BS2}
' {$PBASIC 2.5}
' *****
' * Send_Sleep.bs2 *
' * Illustrates sending a byte every few seconds *
' * while sleeping between transmits. *
' *****
myAddr    CON $1    ' Node Address
DestAddr  CON $0    ' Destination address

Baud      CON 84    ' Baud rate, 9600, 8-N-1, non-inverted, on BS2.

RX        PIN 0     ' Receive Pin
TX        PIN 2     ' Transmit Pin
RTS       PIN 6     ' Flow control Pin
SLP_R     PIN 8     ' Connect to SLP_R

X VAR byte
HIGH TX

DEBUG CLS, "Configuring XBee..."
PAUSE 2000                                ' Guard time for command sequence
SEROUT TX,Baud,["+++"]                    ' Enter command mode
PAUSE 2000                                ' Guard time for command sequence
SEROUT TX,Baud,["ATNI BS2 Test Node",CR,   ' Set description
    "ATMY ", HEX myAddr,CR,              ' Set node address
    "ATDL ", HEX DestAddr,CR,            ' Set destination node address
    "ATD6 1",CR,                          ' Use RTS for flow control
    "ATSM 1",CR,                          ' ***** Sleep mode of 1
    "ATCN",CR]                            ' Exit command mode
PAUSE 1000
DEBUG "Ready!",Cr

DO
HIGH SLP_R                                ' Put XBee to sleep
SLEEP 5                                   ' Put BS2 to sleep for approx 5 seconds
LOW SLP_R                                  ' Wake XBee
PAUSE 10                                   ' Give XBee time to stretch and yawn
x = x + 1                                  ' change a value
SEROUT TX, Baud, [DEC x,CR]               ' Transmit value
DEBUG DEC x,CR                            ' show in DEBUG
PAUSE 10
Loop

```

Sending Data to StampDAQ

Data from multiple units may also send data to a base using (connected to a PC via a USB2SER) and collect that data in StampDAQ from Parallax (developed by SelmaWare Solutions), a serial to Microsoft Excel program. This example shows using a column to denote the node address from which the data is arriving.

```
' {$STAMP BS2}
' {$PBASIC 2.5}
' *****
' * XBee_StampDAQ.bs2 *
' * Illustrates sending a data to StampDAQ *
' *****

myAddr    CON $1    ' Node Address
DestAddr  CON $0    ' Destination address

Baud      CON 84    ' Baud rate, 9600, 8-N-1, non-inverted, on BS2.

RX        PIN 0     ' Receive Pin
TX        PIN 2     ' Transmit Pin
RTS       PIN 6     ' Flow control Pin

X         VAR Byte

HIGH TX

DEBUG CLS, "Configuring XBee..."
PAUSE 2000                                ' Guard time for command sequence
SEROUT TX,Baud,["+++"]                    ' Enter command mode
PAUSE 2000                                ' Guard time for command sequence
SEROUT TX,Baud,["ATNI BS2 Test Node",CR,   ' Set description
               "ATMY ", HEX myAddr,CR,    ' Set node address
               "ATDL ", HEX DestAddr,CR,  ' Set destination node address
               "ATD6 1",CR,              ' Use RTS for flow control
               "ATCN",CR]                ' Exit command mode

PAUSE 1000
DEBUG "Ready!",CR
SEROUT TX,Baud,[CR,"LABEL,TIME,ADDR,X",CR] ' Label 3 TIME, Unit address, X

SEROUT TX,Baud,["CLEARDATA",CR]           ' Clear all data columns (A-J) in Excel

DO
  FOR X = 0 TO 255                         ' Count from 0 to 255
    SEROUT TX,Baud,["DATA,TIME,", DEC myAddr, ",", DEC X, CR] ' Send String with data for Excel
    PAUSE 500                              ' 500 mS wait before next data
  NEXT
LOOP
```

Sending Data to StampPlot

Data may be received into StampPlot from a base unit using a USB2SER and using the Configuration window to select the correct COM port. In this example, the unit's address is used to select the plot channel (0-9) and the color of the plot with the !ACHN instruction of StampPlot:

!ACHN channel, value, color

```
' {$STAMP BS2}
' {$PBASIC 2.5}
' *****
' * XBee_StampPlot.bs2 *
' * Illustrates sending data to StampPlot *
' * via a base unit *
' *****

myAddr    CON $1    ' Node Address
DestAddr  CON $0    ' Destination address

Baud      CON 84    ' Baud rate, 9600, 8-N-1, non-inverted, on BS2.

RX        PIN 0     ' Receive Pin
TX        PIN 2     ' Transmit Pin
RTS       PIN 6     ' Flow control Pin

X VAR Byte
HIGH TX

DEBUG CLS, "Configuring XBee..."
PAUSE 2000                ' Guard time for command sequence
SEROUT TX,Baud,["+++"]   ' Enter command mode
PAUSE 2000                ' Guard time for command sequence
SEROUT TX,Baud,["ATNI BS2 Test Node",CR,    ' Set description
    "ATMY ", HEX myAddr,CR,                ' Set node address
    "ATDL ", HEX DestAddr,CR,              ' Set destination node address
    "ATD6 1",CR,                            ' Use RTS for flow control
    "ATCN",CR]                          ' Exit command mode
PAUSE 1000
DEBUG "Ready!",CR

DO
  FOR X = 0 TO 255                'Count from 0 to 255
    'Send String to StampPlot using channel and color of address (0-9)
    SEROUT TX,Baud,["!ACHN ", DEC myAddr," ", DEC X, " ", DEC myAddr, CR]

  PAUSE 500                '500 mS wait before next data
  NEXT
LOOP
```

Board Construction and Mounting

FAQs

Why do I need an adapter board for the XBee?

The XBee is a 20 pin device with 2mm spacing (breadboards are typically 2.54mm or 0.1 in.). Even if the pins did fit, it would gobble up most of the real estate on smaller breadboards. The XBee is also a 3.3V device. The adapter is regulated for 3.3V and has a 5V to 3.3V buffer. Voltages from the XBee to the controller do not need to be conditioned since they are higher than the BASIC Stamp's threshold voltage (1.8V).

What is IEEE 802.15.4 and ZigBee?

IEEE 802.15.4 is a low-rate communications protocol for wireless networks. The protocol is responsible for moving data between 2 addressable devices, much like Ethernet moves data (IEEE 802.3). This protocol uses the 900MHz or 2.4GHz (XBee) bands, and uses Direct Spread Spectrum Sequence to use very low power for transmission.

ZigBee is a trademark of a consortium of companies in developing compatible applications on top of 802.15.4. ZigBee can be compared to TCP/IP, which uses 802.3 to move data. ZigBee includes the ability to route data between points.

Can I communicate with my WiFi network?

No. IEEE 802.15.4 operates on a different protocol (and purpose!) than WiFi (IEEE 802.11).

What is different about these devices from other RF devices, such as 433MHz ones?

Many RF devices simply send data freely without regard to errors or collisions with data from other devices. IEEE 802.15.4 is a fully implemented protocol ensuring as much as possible that data between devices do not collide (media access control), and the packets arrive without errors. Of course, addressing is a major benefit also.

Does the XBee support routing?

The version of the firmware discussed here, 1.083, is mainly 802.15.4. Other versions are in beta, which may include ZigBee functions of routing, though they may not be compatible with the examples illustrated here. With a little coding, routing may be designed into the BASIC Stamp to assist in these needs.

Can I use other I/O on the XBee that are not pinned out to the header?

Yes, for outputs you may simply run a jumper from the XBee header to your breadboard, keeping in mind it is 3.3V. For inputs to the XBee, your signal must be conditioned using a voltage or other means to 3.3V.

Can I use this board with controllers other than the BASIC Stamp?

Yes, any controller, 5V logic or 3.3V logic may be used, such as the Propeller or SX series from Parallax. The 3.3V regulator on the boards requires at least 5V to V_{in} . If this is unavailable on your 3.3V board, you may customize your board and keep off and jumper the regulator holes pin 1 to 3.

How much current does the AppBee board draw?

With an XBee, current draw is around 50mA. With an XBee-PRO, current draw can exceed 180mA when transmitting. Sleep mode may be used to lower current draw to <1mA, but the unit cannot send nor receive while sleeping.

How can I reprogram the firmware on my XBee?

A serial connection providing DTR and other hardware handshaking lines is required. MaxStream has developer's kits that include boards for development and programming that may be used.

RoHS

Resource Links

Parallax, Inc:

Starter kits: http://www.parallax.com/html_pages/products/kits/starter_kits.asp

USB2SER Device: http://www.parallax.com/detail.asp?product_id=28024

StampDAQ Software:

http://www.parallax.com/html_pages/downloads/software/software_stampDAQ.asp

Other controllers available such as the SX and Propeller multi-core.

MaxStream:

XBee information: <http://www.maxstream.net/products/oem-rf-modules.php>

X-CTU Software: <http://www.maxstream.net/support/downloads.php>

White Papers: <http://www.maxstream.net/support/>

MicroDAQ

XBee devices and development kits (resale):

http://www.microdaq.com/maxstream/zigbee/xbee_development_kits.php

SelmaWare Solutions:

AppBee information: <http://www.selmaware.com/appbee>

StampPlot: <http://www.stampplot.com>